

XML and Semantic Web Technologies

III. Semantic Web / 3. SPARQL Query Language for RDF

Lars Schmidt-Thieme

Information Systems and Machine Learning Lab (ISMLL)
Institute of Economics and Information Systems
& Institute of Computer Science
University of Hildesheim
<http://www.isml.uni-hildesheim.de>

III. Semantic Web / 3. SPARQL Query Language for RDF

1. Basic SPARQL queries

2. Conditional SPARQL Queries

3. Metaqueries/Returning RDF

4. A small integrated example

SPARQL Specification

The SPARQL specification consists of the following parts:

- SPARQL Query Language for RDF (REC-2008/01/15)
- SPARQL Query Results XML Format (REC-2008/01/15)
- SPARQL Protocol for RDF (REC 2008/01/15)

as well as a further document on use cases and requirements.

SPARQL is a query language for RDF that

- has a non-XML syntax,
- makes use of (parts of) XPath as expression language,
- makes use of N3 notation

A simple SPARQL query

```

1 @prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#> .
2 @prefix rdfs: <http://www.w3.org/2000/01/rdf-schema#> .
3 @prefix : <http://www.cgnm.de/rdf/sokrates.rdfs#> .
4 :Mortal rdf:type rdfs:Class .
5 :Human rdf:type rdfs:Class .
6 :Human rdfs:subClassOf :Mortal .
7 :Sokrates rdf:type :Human .
  
```

Figure 1: A sample RDF data file (here in N3, but any notation will do).

```

1 select ?c
2 where { <http://www.cgnm.de/rdf/sokrates.rdfs#Sokrates>
3       <http://www.w3.org/1999/02/22-rdf-syntax-ns#type>
4       ?c }
  
```

Figure 2: A simple SPARQL query.

Executing SPARQL queries

SPARQL is implemented in ARQ, the Jena query processor (by HP).

```
sparql -data sokrates.n3 -query sokrates-simple.sq
```

```

1 -----
2 | c                                     |
3 =====
4 | <http://www.cgnm.de/rdf/sokrates.rdfs#Human> |
5 -----

```

Figure 3: The result of the simple SPARQL query.

SPARQL operates on a RDF graph / set of triples

- explicitly materialized or
- specified implicitly by a an explicitly materialized RDF graph and a set of inference rules.

The ready-to-use commandline tools of ARQ do not support inference yet.

Executing SPARQL queries on Windows

- download ARQ, the Jena query processor, from <http://jena.sourceforge.net/ARQ/> (current version is 2.7.0).
- add environment variable ARQROOT, pointing to the root path where ARQ is extracted to.
- add %ARQROOT%\bat to system path.

Graph Pattern

A SPARQL query basically consists of a query graph pattern (plus some additional information).

The simplest form of a graph pattern is a sequence of triples in N3 notation, i.e.,

syntax	meaning
<code>< <URI> ></code>	relative URI reference
<code>" <string> "</code>	untyped literal
<code>" <string> "^^<URI></code>	typed literal
<code>_: <NCName></code>	anonymous node name
<code><integer></code>	<code>= " <integer> "^^xsd:integer</code>
<code><double></code>	<code>= " <double> "^^xsd:double</code>
<code>true, false</code>	<code>= " true "^^xsd:boolean</code>
<code>? <NCName></code>	variable

Graph Pattern Matching

The basic operation is to retrieve all substitutions of the variables s.t. the query graph pattern after substitution is a subgraph of the source graph.

In SPARQL, URIs could be abbreviated either

- by declaring a namespace prefix and using $\langle QNames \rangle$ (as in N3)
- or by declaring a base URI and using relative URIs.

```

1 prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
2 prefix : <http://www.cgnm.de/rdf/sokrates.rdfs#>
3 select ?c
4 where { :Sokrates rdf:type ?c }

```

Figure 4: Simple SPARQL query using namespace prefixes.

```

1 base <http://www.cgnm.de/rdf/sokrates.rdfs>
2 prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
3 select ?c
4 where { <#Sokrates> rdf:type ?c }

```

Figure 5: Simple SPARQL query using namespace prefixes and a base URI.

SPARQL query syntax

```

<Query> := ( base <QuotedURIref> )?
          ( prefix <NCNAME_PREFIX>? : <QuotedURIref> )*

          ( ask
            | select distinct? ( <Var>+ | * )
            | construct <ConstructTemplate>
            | describe ( ( <Var> | <URI> )+ | * ) )

          ( from named? <URI> )*
          ( where <GraphPattern> )?
          ( order by <OrderCondition>+ )?
          ( limit <INTEGER> )?
          ( offset <INTEGER> )?

```

There is at most one unnamed `from` clause allowed (background graph).
Comments start with `#`.

```

<URI> := <QuotedURIref> | <QName>
<Var> := ( $ | ? ) <StringLiteral>

```

Kinds of SPARQL Queries / Outputs

SPARQL supports $3\frac{1}{2}$ different query types:

1. `ask` returns `true`, if there is at least one substitution, else `false`,
2. `select` returns a the set of substitution tuples (like SQL),
3. `construct` returns a RDF graph build from the substitution tuples and a template (eventually a subgraph of the original graph or a newly constructed graph),
4. `describe` also returns a RDF graph with some implementation-dependent contents.

III. Semantic Web / 3. SPARQL Query Language for RDF

1. Basic SPARQL queries**2. Conditional SPARQL Queries****3. Metaqueries/Returning RDF****4. A small integrated example**

A Fresh Example

```

1 @prefix xs: <http://www.w3.org/2001/XMLSchema#> .
2 @prefix : <http://www.cgnm.de/rdf/family#> .
3 :Anne :age "45"^^xs:integer; :marriedTo :Bert ; :motherOf :Clara, :Dennis .
4 :Bert :age "49"^^xs:integer; :marriedTo :Anne ; :fatherOf :Clara, :Dennis .
5 :Clara :age "24"^^xs:integer; :marriedTo :Emil; :motherOf :Fred, :Gisa .
6 :Dennis :age "22"^^xs:integer.
7 :Emil :age "27"^^xs:integer; :marriedTo :Clara; :fatherOf :Fred, :Gisa .
8 :Fred :age "2"^^xs:integer.
9 :Gisa :age "1"^^xs:integer.

```

Figure 6: A fresh example.

Kinds of Graph Patterns

```

<GraphPattern> := { <PatternElement> ( . <PatternElement> )* }
<PatternElement> := <Triples>
                  | <GraphPattern>
                  | optional? <GraphPattern>
                  | filter <Expression>
                  | <GraphPattern> union <GraphPattern>*
                  | graph ( <Var> | <BlankNode> | <URI> ) <GraphPattern>

```

Optional Graph Patterns

```

1 prefix : <http://www.cgnm.de/rdf/family#>
2 select *
3 where { ?x :marriedTo ?y }

```

x	y
:Emil	:Clara
:Anne	:Bert
:Bert	:Anne
:Clara	:Emil

Figure 7: Query for married persons .

Figure 8: Result.

```

1 prefix : <http://www.cgnm.de/rdf/family#>
2 select *
3 where { ?x :age ?z.
4       optional { ?x :marriedTo ?y } }

```

x	z	y
:Fred	2	
:Emil	27	:Clara
:Bert	49	:Anne
:Anne	45	:Bert
:Clara	24	:Emil
:Dennis	22	
:Gisa	1	

Figure 9: Query for all persons and their spouse (if any).

Constraints / filter

```

1 prefix : <http://www.cgnm.de/rdf/family#>
2 select *
3 where { ?x :age ?z.
4       filter (?z >= 18) }

```

x	z
:Emil	27
:Bert	49
:Anne	45
:Clara	24
:Dennis	22

Figure 11: Query for all adult persons .

Figure 12: Result.

Constraints / filter (2)

Also "negative" queries are possible.

```

1 prefix : <http://www.cgnm.de/rdf/family#>
2
3 SELECT ?s ?age
4 WHERE { ?s :age ?age.
5         optional { ?s :marriedTo ?x}.
6         filter (!bound(?x))
7 }

```

s	age
:Gisa	1
:Fred	2
:Dennis	22

Figure 13: Query for all non-marries persons.

Figure 14: Result.

Unions of Graph Patterns ("group patterns") / union

```

1 prefix : <http://www.cgnm.de/rdf/family#>
2 select *
3 where { { ?x :marriedTo ?y }
4         union { ?x :age ?z.
5               filter (?z < 18) } }

```

x	y	z
:Emil	:Clara	
:Anne	:Bert	
:Bert	:Anne	
:Clara	:Emil	
:Fred		2
:Gisa		1

Figure 15: Query for all married persons and all non-adults.

Figure 16: Result.

Tuples matching several operand graph patterns of a union graph pattern are contained several times in the result.

Restricting returned values (1) / `filter`, `union`

```

1 prefix : <http://www.cgnm.de/rdf/family#>
2 select ?x ?z
3 where {
4   {
5     ?x :age ?z.
6     filter (?z >= 18)
7   } union {
8     ?x :age ?y.
9     filter(?y < 18)
10  }
11 }

```

x	z
:Emil	27
:Dennis	22
:Clara	24
:Bert	49
:Anne	45
:Gisa	
:Fred	

Figure 17: Query for all persons, ignoring age is

Figure 18: Result.

Querying Several Sources / `graph`

SPARQL can deal with several sources at once:

- there is one unnamed default source (**background graph**) and
- there are arbitrary many named further sources (**named graphs**) (where names are specified as URIs)

In the `from` clause (and in implementations), the URI names of a source often are used to locate and retrieve the resource.

E.g., in ARQ named sources can be explicitly given by the `--namedGraph <URI>` commandline option, where the file or http URI is used to retrieve the source.

Querying Several Sources / Example (1/2)

```

1 @prefix xs: <http://www.w3.org/2001/XMLSchema#> .
2 @prefix : <http://www.cgnm.de/rdf/family-miller#> .
3 @prefix r: <http://www.cgnm.de/rdf/relatives#> .
4 :Anne r:age "45"^^xs:integer; r:marriedTo :Bert ; r:motherOf :Clara, :Dennis .
5 :Bert r:age "49"^^xs:integer; r:marriedTo :Anne ; r:fatherOf :Clara, :Dennis .
6 :Clara r:age "24"^^xs:integer; r:marriedTo :Emil; r:motherOf :Fred, :Gisa .
7 :Dennis r:age "22"^^xs:integer.
8 :Emil r:age "27"^^xs:integer; r:marriedTo :Clara; r:fatherOf :Fred, :Gisa .
9 :Fred r:age "2"^^xs:integer.
10 :Gisa r:age "1"^^xs:integer.

```

Figure 19: Data about the Miller family.

```

1 @prefix xs: <http://www.w3.org/2001/XMLSchema#> .
2 @prefix miller: <http://www.cgnm.de/rdf/family-miller#> .
3 @prefix r: <http://www.cgnm.de/rdf/relatives#> .
4 @prefix : <http://www.cgnm.de/rdf/family-smith#> .
5 :Adam r:age "52"^^xs:integer; r:marriedTo :Britta ; r:motherOf :Emil .
6 :Emil r:age "27"^^xs:integer; r:marriedTo miller:Clara; r:fatherOf miller:Fred, miller:G

```

Figure 20: Data about the Smith family.

Lars Schmidt-Thieme, Information Systems and Machine Learning Lab (ISMLL), University of Hildesheim, Germany,
Course on XML and Semantic Web Technologies, summer term 2009

18/45

Querying Several Sources / Example (2/2)

```

1 prefix : <http://www.cgnm.de/rdf/relatives#>
2 prefix miller: <http://www.cgnm.de/rdf/family-miller#>
3 prefix smith: <http://www.cgnm.de/rdf/family-smith#>
4 select *
5 where {
6   { ?x :marriedTo ?y }
7   union
8   { graph <file:family-smith.n3>
9     { ?x :marriedTo ?y } }
10 }

```

Figure 21: Query for all married people in these two families.

x	y
miller:Clara	miller:Emil
miller:Anne	miller:Bert
miller:Bert	miller:Anne
miller:Emil	miller:Clara
smith:Emil	miller:Clara
smith:Adam	smith:Britta

Figure 22: Result.

Lars Schmidt-Thieme, Information Systems and Machine Learning Lab (ISMLL), University of Hildesheim, Germany,
Course on XML and Semantic Web Technologies, summer term 2009

19/45

Specifying Data Sources / `from` clause

`from` may be used to explicitly point to a target resource URI.

```

1 prefix : <http://www.cgnm.de/rdf/family-miller#>
2 prefix r: <http://www.cgnm.de/rdf/relatives#>
3
4 SELECT *
5 FROM <file:family-miller.n3>
6 WHERE {
7     ?s r:age ?age. FILTER(?age >30)
8 }
```

Figure 23: Sample Query for `from`.

```

1 -----
2 | s      | age |
3 =====
4 | :Bert  | 49  |
5 | :Anne  | 45  |
6 -----
```

Figure 24: Result.

Specifying Data Sources / `from named` clause

`from named` may be used to use arbitrarily many background graphs in the query. URIs are resolved against the namespace declarations in the query.

```

1 prefix r: <http://www.cgnm.de/rdf/relatives#>
2 prefix miller: <http://www.cgnm.de/rdf/family-miller#>
3 prefix smith: <http://www.cgnm.de/rdf/family-smith#>
4 SELECT ?age ?s
5 FROM NAMED <file:family-smith.n3>
6 WHERE {
7     graph <file:family-smith.n3> {?s r:age ?age. FILTER(?age >30)}
8 }
```

Figure 25: Querying the ages of persons in the smith family.

```

1 -----
2 | age | s      |
3 =====
4 | 52  | smith:Adam |
5 -----
```

Figure 26: Result.

Specifying Data Sources / `from named` clause (2)

`from named` may be used to use arbitrarily many background graphs in the query. URIs are resolved against the namespace declarations in the query.

```

1 prefix r: <http://www.cgnm.de/rdf/relatives#>
2 prefix miller: <http://www.cgnm.de/rdf/family-miller#>
3 prefix smith: <http://www.cgnm.de/rdf/family-smith#>
4 SELECT ?g ?age ?s
5 FROM NAMED <file:family-smith.n3>
6 FROM NAMED <file:family-miller.n3>
7 WHERE {
8     graph ?g {?s r:age ?age. FILTER(?age >30)}
9 }

```

Figure 27: Querying the ages of persons in different background graphs.

```

1 -----
2 | g                | age | s                |
3 =====
4 | <family-smith.n3> | 52  | smith:Adam       |
5 | <family-miller.n3> | 49  | miller:Bert      |
6 | <family-miller.n3> | 45  | miller:Anne      |
7 -----

```

Figure 28: Result.

Lars Schmidt-Thieme, Information Systems and Machine Learning Lab (ISMLL), University of Hildesheim, Germany, Course on XML and Semantic Web Technologies, summer term 2009

22/45

Querying implicit Data Sources

One Drawback of `from/ from named` is that background graphs need to be known at query writing time.

This holds only true in very rare situations.

It is (e.g. with ARQ) easily possible, to dynamically add (named) background graphs ("context"s) using the command line.

Querying implicit Data Sources / Example

```
sparql --namedGraph family-miller.n3 --query from-implicit-named.sq
```

```
1 prefix r: <http://www.cgnm.de/rdf/relatives#>
2 prefix miller: <http://www.cgnm.de/rdf/family-miller#>
3 prefix smith: <http://www.cgnm.de/rdf/family-smith#>
4 SELECT ?g ?age ?s
5 WHERE {
6     graph ?g {?s r:age ?age}.
7     FILTER(?age >30)
8 }
```

Figure 29: Querying the ages of persons in unknown named background graphs.

```
1 -----
2 | g                | age | s                |
3 =====
4 | <family-miller.n3> | 49  | miller:Bert      |
5 | <family-miller.n3> | 45  | miller:Anne     |
6 -----
```

Figure 30: Result.

Querying implicit Data Sources / Example (2)

```
sparql --namedGraph family-miller.n3 --namedGraph family-smith.n3 --query
from-implicit-named.sq
```

```
1 prefix r: <http://www.cgnm.de/rdf/relatives#>
2 prefix miller: <http://www.cgnm.de/rdf/family-miller#>
3 prefix smith: <http://www.cgnm.de/rdf/family-smith#>
4 SELECT ?g ?age ?s
5 WHERE {
6     graph ?g {?s r:age ?age}.
7     FILTER(?age >30)
8 }
```

Figure 31: Querying the ages of persons in unknown named background graphs.

```
1 -----
2 | g                | age | s                |
3 =====
4 | <family-smith.n3>  | 52  | smith:Adam      |
5 | <family-miller.n3> | 49  | miller:Bert     |
6 | <family-miller.n3> | 45  | miller:Anne     |
7 -----
```

Figure 32: Result.

Querying implicit Data Sources / Example (3)

But: If triples are loaded as a background graph, they have to be queried from that location.

```
sparql --namedGraph family-miller.n3 --query from-implicit-named.sq
```

```
1 prefix r: <http://www.cgnm.de/rdf/relatives#>
2 SELECT ?age ?s
3 WHERE {
4     ?s r:age ?age. FILTER(?age >30)
5 }
```

Figure 33: Querying the ages of persons in the default background graph.

```
1 -----
2 | age | s |
3 =====
4 -----
```

Figure 34: Result.

Sorting / `order` clause
$$\langle \text{OrderCondition} \rangle := \langle \text{FunctionCall} \rangle \mid \langle \text{Var} \rangle$$

$$\mid (\text{asc} \mid \text{desc}) ((\langle \text{FunctionCall} \rangle \mid \langle \text{Var} \rangle))$$

```
1 prefix : <http://www.cgnm.de/rdf/family#>
2 select *
3 where { ?x :age ?z }
4 order by desc (?z) desc (?x)
5
6 | :Bert      | 49 |
7 | :Anne      | 45 |
8 | :Emil      | 27 |
9 | :Clara     | 24 |
10 | :Dennis    | 22 |
11 | :Fred      | 2  |
12 | :Gisa      | 1  |
```

Figure 35: Query for all persons sorted by descending age and ascending URI.

Figure 36: Result.

Simple Cursor Functionalities / `limit` and `offset` clause

```

1 prefix : <http://www.cgnm.de/rdf/family#>
2 select *
3 where { ?x :age ?z }
4 order by desc (?x)
5 limit 2
6 offset 3

```

x	z
:Clara	24
:Dennis	22

Figure 37: Query for a subset of all persons sorted by descending age and ascending URI. Figure 38: Result.

III. Semantic Web / 3. SPARQL Query Language for RDF

1. Basic SPARQL queries

2. Conditional SPARQL Queries

3. Metaqueries/Returning RDF

4. A small integrated example

ask - Querying Statement existence

It is possible to ask whether statement(s) in a source graph exist.

```
1 prefix : <http://www.cgnm.de/rdf/family#>
```

```
2
3 ask {
```

```
4   :Emil :fatherOf :Fred.
```

```
5 }
```

```
. Ask => Yes
```

Figure 39: Is Emil the father of Fred?

Figure 40: Result.

ask - a more complex example

```
1 prefix : <http://www.cgnm.de/rdf/family#>
```

```
2
3 ask {
```

```
4   ?x :fatherOf ?y.
```

```
5   ?x :age ?age.
```

```
6   FILTER (?age <= 18).
```

```
7 }
```

```
. Ask => No
```

Figure 41: A more complex example.

Figure 42: Result.

Selecting Subgraphs of the Source

It is possible to copy a whole source on basis of its triples.

```
select * where { ?s ?p ?o }
```

Figure 43: Query for all triples in the source.

In the same manner, subsets of triples meeting some conditions can be selected, resulting in a subgraph of the source.

Creating New Triples

New triples can be created by the `construct` query, a graph template that contains only triples and variables (but no `optional`, `graph` etc. statements).

$$\langle \text{ConstructTemplate} \rangle := \{ \langle \text{Triples} \rangle (. \langle \text{Triples} \rangle)^* . ? \}$$

The template is instantiated once for each result tuple, whereat variables are substituted by the values of result tuples.

Example

```

1 prefix : <http://www.cgnm.de/rdf/family#>
2 prefix r: <http://www.cgnm.de/rdf/relatives#>
3 construct {
4   ?x r:marriedTo ?spouse .
5   ?x r:isMarried true }
6 where { ?x :marriedTo ?spouse }

```

Figure 44: Query that recodes the marriedOf property.

```

1 @prefix r: <http://www.cgnm.de/rdf/relatives#>
2 @prefix xs: <http://www.w3.org/2001/XMLSchema#>
3 @prefix : <http://www.cgnm.de/rdf/family#>
4
5 :Anne r:isMarried "true"^^xs:boolean ;
6       r:marriedTo :Bert .
7 :Emil r:isMarried "true"^^xs:boolean ;
8       r:marriedTo :Clara .
9 :Clara r:isMarried "true"^^xs:boolean ;
10      r:marriedTo :Emil .
11 :Bert r:isMarried "true"^^xs:boolean ;
12      r:marriedTo :Anne .

```

Figure 45: Result.

Example (2)

```

1 prefix ismll: <http://www.ismll.de/rdf/meta#>
2 prefix r: <http://www.cgnm.de/rdf/relatives#>
3 prefix miller: <http://www.cgnm.de/rdf/family-miller#>
4 prefix smith: <http://www.cgnm.de/rdf/family-smith#>
5
6 construct {
7   ?x r:age ?age .
8   ?x ismll:source ?g.
9 }
10 where {
11   graph ?g {?x r:age ?age.}
12 }
13
14 @prefix r: <http://www.cgnm.de/rdf/relatives#>
15 @prefix miller: <http://www.cgnm.de/rdf/family-miller#>
16 @prefix smith: <http://www.cgnm.de/rdf/family-smith#>
17 @prefix ismll: <http://www.ismll.de/rdf/meta#>
18
19 smith:Adam
20   r:age      52 ;
21   ismll:source <file:///C:/Users/busche/v...>
22
23 miller:Anne
24   r:age      45 ;
25   ismll:source <file:///C:/Users/busche/v...>
26
27 miller:Gisa
28   r:age      1 ;
29   ismll:source <file:///C:/Users/busche/v...>

```

Figure 46: Adding source information.

```

17 smith:Emil
18 miller:Dennis
19   r:age      22 ;
20   ismll:source <file:///C:/Users/busche/workspace35_2/XML2009_script/skript/ex>
21
22 miller:Emil
23   r:age      27 ;
24   ismll:source <file:///C:/Users/busche/workspace35_2/XML2009_script/skript/ex>
25
26 miller:Fred
27   r:age      2 ;
28   ismll:source <file:///C:/Users/busche/workspace35_2/XML2009_script/skript/ex>
29
30 smith:Emil
31   r:age      27 ;
32   ismll:source <file:///C:/Users/busche/v
33
34 miller:Clara
35   r:age      24 ;
36   ismll:source <file:///C:/Users/busche/v
37
38 miller:Bert
39   r:age      49 ;
40   ismll:source <file:///C:/Users/busche/v

```

Figure 47: Result.

Describe - returning all information about selected resources

Describes a (set of) resources in terms of subject, predicate, and object .

```

1 prefix : <http://www.cgnm.de/rdf/family#>

```

```

2
3 describe :Anne

```

Figure 48: Query describing Anne.

```

1 @prefix r:      <http://www.cgnm.de/rdf/relatives#> .
2 @prefix xs:    <http://www.w3.org/2001/XMLSchema#> .
3 @prefix :      <http://www.cgnm.de/rdf/family#> .
4
5 :Anne
6   :age      45 ;
7   :marriedTo :Bert ;
8   :motherOf :Clara ;
9   :motherOf :Dennis .

```

Figure 49: Result.

Describe, Example (2)

```

1 prefix : <http://www.cgnm.de/rdf/family#>. @prefix xs: <http://www.w3.org/2001/XMLSchema#>
2 @prefix : <http://www.cgnm.de/rdf/family#>.
3 describe ?x ?y {
4   ?x :age ?age.
5   FILTER (?age >= 18).
6   ?x :fatherOf ?y.
7 }

```

Figure 50: Query describing fathers.

```

8 :Fred
9   :age 27 ;
10  :fatherOf :Fred ;
11  :fatherOf :Gisa ;
12  :marriedTo :Clara .
13 :Clara
14  :age 24 ;
15  :marriedTo :Emil ;
16  :motherOf :Fred ;
17  :motherOf :Gisa .
18 :Gisa
19  :age 1 .
20 :Dennis
21  :age 22 .
22 :Bert
23  :age 49 ;
24  :fatherOf :Clara .

```

III. Semantic Web / 3. SPARQL Query Language for RDF

1. Basic SPARQL queries**2. Conditional SPARQL Queries****3. Metaqueries/Returning RDF****4. A small integrated example**

FOAF - Friend of a Friend

The FOAF Project was founded to describe people, and their links, through a (standardized) vocabulary.

The FOAF Specification consists of its namespace document:

- FOAF Vocabulary Specification 0.91 (SPEC-2007/11/02)

available at <http://xmlns.com/foaf/spec/>.

FOAF / Example

```

1 <rdf:RDF xmlns:rdf="http://www.w3.org/1999/02/22-rdf-syntax-ns#"
2   xmlns:rdfs="http://www.w3.org/2000/01/rdf-schema#"
3   xmlns:foaf="http://xmlns.com/foaf/0.1/">
4   <foaf:Person rdf:ID="me">
5     <foaf:name>Lars Schmidt-Thieme</foaf:name>
6     <foaf:title>Herr Prof. Dr. Dr.</foaf:title>
7     <foaf:givenname>Lars</foaf:givenname>
8     <foaf:family_name>Schmidt-Thieme</foaf:family_name>
9     <foaf:mbox_sha1sum>3ff97691d04aa172889553373a3467666bf7c100</foaf:
10    </foaf:Person>
11 </rdf:RDF>

```

Figure 52: Very basic information.

Combining FOAF with our example

```

1 @prefix miller: <http://www.cgnm.de/rdf/family-miller#> .
2 @prefix smith: <http://www.cgnm.de/rdf/family-smith#> .
3 @prefix rdf:    <http://www.w3.org/1999/02/22-rdf-syntax-ns#> .
4 @prefix foaf:  <http://xmlns.com/foaf/0.1/> .
5
6 smith:Emil foaf:name "Emil".
7 smith:Emil rdf:type foaf:Person.
8
9 miller:Clara foaf:name "Clara".
10 miller:Clara rdf:type foaf:Person.
11
12 smith:Emil foaf:knows miller:Clara.
  
```

Figure 53: Adding FOAF-Statements to our example.

Simple queries (1)

```

1 prefix :          <http://www.cgnm.de/rdf/relatives#>
2 prefix miller:   <http://www.cgnm.de/rdf/family-miller#>
3 prefix smith:    <http://www.cgnm.de/rdf/family-smith#>
4 prefix foaf:     <http://xmlns.com/foaf/0.1/>
5 prefix rdf:      <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
6
7 SELECT *
8 where {
9   ?x rdf:type foaf:Person.
10 }
  
```

Figure 54: Query for all people.

```

1 -----
2 | x                |
3 =====
4 | miller:Clara    |
5 | smith:Emil      |
6 -----
  
```

Figure 55: Result.

Simple queries (2)

```

1 prefix :           <http://www.cgnm.de/rdf/relatives#>
2 prefix miller:    <http://www.cgnm.de/rdf/family-miller#>
3 prefix smith:     <http://www.cgnm.de/rdf/family-smith#>
4 prefix foaf:      <http://xmlns.com/foaf/0.1/>
5 prefix rdf:       <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
6
7 SELECT *
8 where {
9   ?x rdf:type foaf:Person.
10  graph <file:family-smith.n3>
11    { ?x :marriedTo ?y}
12 }

```

Figure 56: Query for all people being married.

```

1 -----
2 | x           | y
3 =====
4 | smith:Emil | miller:Clara |
5 -----

```

Figure 57: Result.

Lars Schmidt-Thieme, Information Systems and Machine Learning Lab (ISMLL), University of Hildesheim, Germany,
Course on XML and Semantic Web Technologies, summer term 2009

42/45

Simple queries (3)

```

1 prefix :           <http://www.cgnm.de/rdf/relatives#>
2 prefix miller:    <http://www.cgnm.de/rdf/family-miller#>
3 prefix smith:     <http://www.cgnm.de/rdf/family-smith#>
4 prefix foaf:      <http://xmlns.com/foaf/0.1/>
5 prefix rdf:       <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
6
7 SELECT * where {
8   ?x rdf:type foaf:Person.
9   graph <file:family-smith.n3>
10    { ?x :marriedTo ?y}
11   graph <file:family-miller.n3>
12    { ?y :marriedTo ?z.
13      ?y :age ?y_age.} }

```

Figure 58: Symmetric properties? Inferencing?

```

1 -----
2 | x           | y           | z           | y_age |
3 =====
4 | smith:Emil | miller:Clara | miller:Emil | 24    |
5 -----

```

Figure 59: Result.

Lars Schmidt-Thieme, Information Systems and Machine Learning Lab (ISMLL), University of Hildesheim, Germany,
Course on XML and Semantic Web Technologies, summer term 2009

43/45

Simple queries (3)

It is impossible with RDF(S) to define properties that define aspects like symmetry, identity, etc.

```

1 prefix : <http://www.cgnm.de/rdf/relatives#>
2 prefix miller: <http://www.cgnm.de/rdf/family-miller#>
3 prefix smith: <http://www.cgnm.de/rdf/family-smith#>
4 prefix foaf: <http://xmlns.com/foaf/0.1/>
5 prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

```

```

6
7 SELECT * where {
8   ?x rdf:type foaf:Person.
9   ?z rdf:type foaf:Person.
10  graph <file:family-smith.n3> { ?x :marriedTo ?y}
11  graph <file:family-miller.n3> { ?y :marriedTo ?z.
12                                     ?y :age ?y_age.} }

```

Figure 60: No inferencing.

```

1 -----
2 | x | z | y | y_age |
3 =====
4 -----

```

Figure 61: Result.

Lars Schmidt-Thieme, Information Systems and Machine Learning Lab (ISMLL), University of Hildesheim, Germany,
Course on XML and Semantic Web Technologies, summer term 2009